

Dissecting Different Flavors of Generic Programming in One Haskell Universe

Presented to Galois

Sean Leather

Utrecht University

August 27, 2013

What is Generic Programming?

In programming languages, the adjective “generic” is heavily overloaded.

- Java/C# generics
- C++ templates
- Ada generic packages

What is Generic Programming?

The goal is often the same.

A higher level of abstraction than “normally” available

The technique is also often similar.

Some form of parameterization and instantiation

Examples of Generic Programming

Java/C#:

```
public class Stack<T>
{
    public void push(T item) {...}
    public T pop() {...}
}
```

In other words:

- Java-style generics \approx parametric polymorphism

Examples of Generic Programming

C++:

```
template<typename T, typename Compare>
T& min(T& a, T& b, Compare comp) {
    if (comp(b, a))
        return b;
    return a;
}
```

In other words:

- C++ templates \approx ad-hoc polymorphism

Generic Programming in Haskell

“Generic programming”:

- For other languages, the term tends to be used for late additions.
- Parametric and ad-hoc polymorphism were available in Haskell from the beginning.

In Haskell, we have come to use “generic programming” for **datatype-generic programming** (a.k.a. “polytypism” or “shape/structure polymorphism”).

Datatype-Generic Programming

What is datatype-generic programming?

- Parameterize a function over the *structure* of datatypes
- Instantiate the function with a particular type

The result is a function that

- **works with many types** (polymorphism) but
- **uses knowledge of the type** (unlike parametric) and
- **need not be redefined for every type** (unlike ad-hoc).

Generic Functions

Applications

- Pretty-printing (e.g. `show`), parsing (e.g. `read`)
- Compression, serialization, marshallng (and their inverses)
- Comparison, equality
- (Co-)recursion, map, zip, zippers
- Traversals, queries, updates

Generic Platforms

Many different implementations:

- Preprocessors:
 - ▶ PolyP
 - ▶ Generic Haskell
- Libraries
 - ▶ Scrap Your Boilerplate (SYB) – included with GHC for a long time
 - ▶ Extensible and Modular Generics for the Masses (EMGM)
 - ▶ Regular – recursion schemes
 - ▶ Multirec – mutually recursive datatypes
 - ▶ Generic Deriving – available in GHC ≥ 7.2 , similar to Instant Generics
 - ▶ (and many, many more)

Generic Flavors

The implementations can be grouped into flavors depending on how they view the structure of a datatype.

Some flavors (or views):

Spine A constructor is a sequence of types.

Example: SYB

Sums-of-products A datatype is a collection of alternative tuples of types.

Example: Generic Deriving

Fixed point A datatype is a sums-of-products with recursive points.

Example: Multirec

Dissecting a Datatype: Sums-of-Products

```
data Tsum = A1 | A2
```

A datatype can have:

- **Alternatives:** unique constructors (≥ 0)

Dissecting a Datatype: Sums-of-Products

```
data Tprod = P2 Char Int
```

A datatype can have:

- **Fields**: types for each constructor (≥ 0)

Dissecting a Datatype: Sums-of-Products

Other features that are modeled:

- Constant types: each type in a field
- Parameters: type variables (≥ 0)

Features that are not modeled:

- Recursion
- Nesting (though it can be)

Modeling a Sum

To model (nested) alternatives:

```
data Either a b = Left a | Right b
```

For syntactic elegance:

```
data a :+: b = L a | R b
```

Modeling a Product

To model (nested) fields:

```
data (,) a b = (,) a b
```

For syntactic elegance:

```
data a :×: b = a :×: b
```

Modeling Other Structures

A constructor without fields:

```
data U = U
```

A constructor name:

```
data C a = C String a
```

A field type:

```
data K a = K a
```

Note: There are other features of datatypes, but we consider only the above.

Modeling an Example

An example datatype:

```
data E a = E1 | E2 a (E a) Int
```

The corresponding **structure representation type**:

```
type RepE a = C U :+: C (K a :×: K (E a) :×: K Int)
```

Notes:

- `:+:` is **infixr 5** and `:×:` is **infixr 6**.
- Operators nest to the right.

Converting Between Types: Isomorphism

- Generic functions work on the sums-of-products model.
- But first we need to convert between the model and the actual value of the datatype.
- We define an **isomorphism**: two total, dual functions.

```
fromE :: E a → RepE a
fromE E1           = L (C "E1" U)
fromE (E2 x e i) = R (C "E2" ((K x) :×: (K e) :×: (K i)))
```

```
toE :: RepE a → E a
toE (L (C "E1" U))           = E1
toE (R (C "E2" ((K x) :×: (K e) :×: (K i)))) = E2 x e i
```

Converting Between Types: Isomorphism

For convenience, we join the representation type and isomorphism in a type class `Generic` with an associated type synonym `Rep`.

```
class Generic a where  
  type Rep a  
  from :: a → Rep a  
  to   :: Rep a → a
```

The instance for `E`:

```
instance Generic (E a) where  
  type Rep (E a) = RepE a  
  from = fromE  
  to   = toE
```

Generic Functions

A **generic function**

- Is defined on each case of the structure representation and
- Works for every datatype that has a structure representation and isomorphism.

Example: `showRep a :: a → String`

- We will define a `show` function for each case.

Defining a Generic Function: `show`

Unit:

```
showU :: U → String  
showU U = ""
```

Constructor name:

```
showC :: (a → String) → C a → String  
showC showa (C nm a) = "(" ++ nm ++ " " ++ showa a ++ ")"
```

Field:

```
showK :: (a → String) → K a → String  
showK showa (K a) = showa a
```

Defining a Generic Function: `show`

Binary sum:

```
show+ :: (a → String) → (b → String) → a :+: b → String
show+ showa (L a) = showa a
show+ showb (R b) = showb b
```

Binary product:

```
show× :: (a → String) → (b → String) → a :×: b → String
show× showa showb (a :×: b) = showa a ++ " " ++ showb b
```

Defining a Generic Function: `show`

Recall:

```
type RepE a = C U :+: C (K a :×: K (E a) :×: K Int)
```

We can define a `show` function (assuming `showInt`):

```
showRepE :: (a → String) → ((a → String) → E a → String)  
           → RepE a → String
```

```
showRepE showa showE =  
  show+ (showC showU)  
        (showC (show× (showK showa)  
                      (show× (showK (showE showa)) (showK showInt))))))
```

Defining a Generic Function: `show`

```
showRepE :: (a → String) → ((a → String) → E a → String)
              → RepE a → String

showRepE showa showE =
  show+ (showC showU)
        (showC (show× (showK showa)
                      (show× (showK (showE showa)) (showK showInt))))))
```

The `showE` function itself is just an isomorphism away:

```
showE :: (a → String) → E a → String
showE showa = showRepE showa showE ∘ fromE
```


Defining a Generic Function: `show`

```
showRepE :: (a → String) → ((a → String) → E a → String)
              → RepE a → String

showRepE showa showE =
  show+ (showC showU)
        (showC (show× (showK showa)
                      (show× (showK (showE showa)) (showK showInt))))))
```

Some observations:

- This is **not** a generic function.
- It is defined on the structure of `E`, not on datatypes in general.
- It demonstrates a predictable pattern for defining the generic function.

Defining a Generic Function: `show`

Consider these typical expressions and their types:

```
showC showU :: C U → String
show× (showK showInt) (showK showChar) :: (K Int :×: K Char) → String
```

- `show?` functions call other `show?` functions.
- They can be considered recursive but not in the usual way.
- **Polymorphic recursion** – functions with different types that have a common scheme that reference each other

Defining a Generic Function: `show`

There are several ways to encode polymorphic recursion. We use type classes.

- The class declaration specifies the type signature.
- Each recursive (type) case is specified by an instance of the class.

A simplified definition of the `Show` class:

```
class Show a where  
  show :: a → String
```

Defining a Generic Function: `show`

Some of the instances for each structure representation case:

Constructor name:

```
instance Show a  $\Rightarrow$  Show (C a) where  
  show = showC show
```

Binary sum:

```
instance (Show a, Show b)  $\Rightarrow$  Show (a :+: b) where  
  show = show+ show show
```

The remaining instances are straightforward.

Defining a Generic Function: `show`

Now, compare:

```
showRepE :: (a → String) → ((a → String) → E a → String)
              → RepE a → String

showRepE showa showE =
  show+ (showC showU)
        (showC (show× (showK showa)
                      (show× (showK (showE showa)) (showK showInt))))))
```

To:

```
showRepE :: (Show a, Show (E a)) ⇒ RepE a → String
showRepE = show
```

Defining a Generic Function: `show`

Finally, we can use a slightly different `Show` class to support generic functions for any type that has a representation.

class `Show a where`

`show :: a → String`

default `show :: (Show (Rep a), Generic a) ⇒ a → String`

`show = show ∘ from`

- This uses default signatures: if type `a` has the instances `Show (Rep a)` and `Generic a`, then the given definition is used.

The instance for `E`:

instance `Show a ⇒ Show (E a)`

???

class Uniplate' a r **where**

descend' :: (r → r) → a → a

instance Uniplate' U a **where**

descend' _ U = U

instance Uniplate a ⇒ Uniplate' (K a) a **where**

descend' f (K a) = K (f a)

instance Uniplate' (K a) r **where**

descend' _ (K a) = K a

instance Uniplate' a r ⇒ Uniplate' (C a) r **where**

descend' f (C nm a) = C nm (descend' f a)

instance (Uniplate' a r, Uniplate' b r) ⇒ Uniplate' (a :+: b) r **where**

descend' f (L a) = L (descend' f a)

descend' f (R b) = R (descend' f b)

instance (Uniplate' a r, Uniplate' b r) ⇒ Uniplate' (a :×: b) r **where**

descend' f (a :×: b) = descend' f a :×: descend' f b

GP in General

- Datatype-generic programming:
 - ▶ Datatype is the parameter
 - ▶ Instantiation gives you a large class of generic functions
- Many generic functions:
 - ▶ Pretty-printing (`show`) and parsing (`read`)
 - ▶ Compression, serialization, and the reverse
 - ▶ Comparison, equality
 - ▶ Folds (catamorphisms), unfolds (anamorphisms), maps, zips, zippers
 - ▶ Traversals, updates, queries
- Many different libraries:
 - ▶ Instant Generics – presented here
 - ▶ Generic Deriving – GHC \geq 7.2, similar to Instant Generics
 - ▶ EMGM – maintained by me
 - ▶ Regular – folds, etc.
 - ▶ Multirec – mutually recursive datatypes, folds, etc.
 - ▶ Scrap Your Boilerplate (SYB) – GHC, traversals, queries
 - ▶ ...

References

Generic Programming in Haskell:

- Johan Jeuring, Sean Leather, José Pedro Magalhães, Alexey Rodriguez Yakushev. *Libraries for Generic Programming in Haskell*. AFP 2008. pp. 165-229, 2009.
- Generic Deriving:
<http://www.haskell.org/haskellwiki/GHC.Generics>